

# Verification of the Deutsch-Schorr-Waite Graph Marking Algorithm using Data Refinement

Viorel Preoteasa and Ralph-Johan Back

June 18, 2021

## Abstract

The verification of the Deutsch-Schorr-Waite graph marking algorithm is used as a benchmark in many formalizations of pointer programs. The main purpose of this mechanization is to show how data refinement of invariant based programs can be used in verifying practical algorithms. The verification starts with an abstract algorithm working on a graph given by a relation *next* on nodes. Gradually the abstract program is refined into Deutsch-Schorr-Waite graph marking algorithm where only one bit per graph node of additional memory is used for marking.

## Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Address Graph</b>	<b>3</b>
<b>3</b>	<b>Marking Using a Set</b>	<b>5</b>
3.1	Transitions . . . . .	6
3.2	Invariants . . . . .	6
3.3	Diagram . . . . .	7
3.4	Correctness of the transitions . . . . .	8
3.5	Diagram correctness . . . . .	8
<b>4</b>	<b>Marking Using a Stack</b>	<b>9</b>
4.1	Transitions . . . . .	9
4.2	Invariants . . . . .	10
4.3	Data refinement relations . . . . .	10
4.4	Data refinement of the transitions . . . . .	11
4.5	Diagram data refinement . . . . .	11
4.6	Diagram correctness . . . . .	11

<b>5</b>	<b>Generalization of Deutsch-Schorr-Waite Algorithm</b>	<b>11</b>
5.1	Transitions . . . . .	14
5.2	Invariants . . . . .	14
5.3	Data refinement relations . . . . .	14
5.4	Diagram . . . . .	15
5.5	Data refinement of the transitions . . . . .	15
5.6	Diagram data refinement . . . . .	16
5.7	Diagram correctness . . . . .	16
<b>6</b>	<b>Deutsch-Schorr-Waite Marking Algorithm</b>	<b>17</b>
6.1	Transitions . . . . .	18
<b>7</b>	<b>Data refinement relation</b>	<b>19</b>
7.1	Data refinement of the transitions . . . . .	19
7.2	Diagram data refinement . . . . .	20
7.3	Diagram corectness . . . . .	20

## 1 Introduction

The verification of the Deutsch-Schorr-Waite (DSW) [14, 10] graph marking algorithm is used as a benchmark in many formalizations of pointer programs [11, 1]. The main purpose of this mechanization is to show how data refinement [12] of invariant based programs [3, 4, 5, 6] can be used in verifying practical algorithms.

The DSW algorithm marks all nodes in a graph that are reachable from a *root* node. The marking is achieved using only one extra bit of memory for every node. The graph is given by two pointer functions, *left* and *right*, which for any given node return its left and right successors, respectively. While marking, the left and right functions are altered to represent a stack that describes the path from the root to the current node in the graph. On completion the original graph structure is restored. We construct the DSW algorithm by a sequence of three successive data refinement steps. One step in these refinements is a generalization of the DSW algorithm to an algorithm which marks a graph given by a family of pointer functions instead of left and right only.

Invariant based programming is an approach to construct correct programs where we start by identifying all basic situations (pre- and post-conditions, and loop invariants) that could arise during the execution of the algorithm. These situations are determined and described before any code is written. After that, we identify the transitions between the situations, which together determine the flow of control in the program. The transitions are verified at the same time as they are constructed. The correctness of the program is thus established as part of the construction process.

Data refinement [9, 2, 7, 8] is a technique of building correct programs working on concrete data structures as refinements of more abstract programs working on abstract data structures. The correctness of the final program follows from the correctness of the abstract program and from the correctness of the data refinement.

Both the semantics and the data refinement of invariant based programs were formalized in [13], and this verification is based on them.

We use a simple model of pointers where addresses (pointers, nodes) are the elements of a set and pointer fields are global pointer functions from addresses to addresses. Pointer updates ( $x.left := y$ ) are done by modifying the global pointer function  $left := left(x := y)$ . Because of the nature of the marking algorithm where no allocation and disposal of memory are needed we do not treat these operations.

A number of Isabelle techniques are used here. The class mechanism is used for extending the complete lattice theories as well as for introducing well founded and transitive relations. The polymorphism is used for the state of the computation. In [13] the state of computation was introduced as a type variable, or even more generally, state predicates were introduced as elements of a complete (boolean) lattice. Here the state of the computation is instantiated with various tuples ranging from the abstract data in the first algorithm to the concrete data in the final refinement. The locale mechanism of Isabelle is used to introduce the specification variables and their invariants. These specification variables are used for example to prove that the main variables are restored to their initial values when the algorithm terminates. The locale extension and partial instantiation mechanisms turn out to be also very useful in the data refinements of DSW. We start with a locale which fixes the abstract graph as a relation *next* on nodes. This locale is first partially interpreted into a locale which replaces *next* by a union of a family of pointer functions. In the final refinement step the locale of the pointer functions is interpreted into a locale with only two pointer functions, *left* and *right*.

## 2 Address Graph

```
theory Graph
imports Main
begin
```

This theory introduces the graph to be marked as a relation *next* on nodes (addresses). We assume that we have a special node *nil* (the null address). We have a node *root* from which we start marking the graph. We also assume that *nil* is not related by *next* to any node and any node is not related by *next* to *nil*.

```
locale node =
```

```

fixes nil    :: 'node
fixes root  :: 'node

locale graph = node +
  fixes next :: ('node × 'node) set
  assumes next-not-nil-left: (!! x . (nil, x) ∉ next)
  assumes next-not-nil-right: (!! x . (x, nil) ∉ next)
begin

```

On lists of nodes we introduce two operations similar to existing `hd` and `tl` for getting the head and the tail of a list. The new function `head` applied to a nonempty list returns the head of the list, and it returns `nil` when applied to the empty list. The function `tail` returns the tail of the list when applied to a non-empty list, and it returns the empty list otherwise.

**definition**

$head\ S \equiv (if\ S = []\ then\ nil\ else\ (hd\ S))$

**definition**

$tail\ (S::'a\ list) \equiv (if\ S = []\ then\ []\ else\ (tl\ S))$

**lemma** [simp]:  $((nil, x) \in next) = False$   
 ⟨proof⟩

**lemma** [simp]:  $((x, nil) \in next) = False$   
 ⟨proof⟩

**theorem** *head-not-nil* [simp]:

$(head\ S \neq nil) = (head\ S = hd\ S \wedge tail\ S = tl\ S \wedge hd\ S \neq nil \wedge S \neq [])$   
 ⟨proof⟩

**theorem** *nonempty-head* [simp]:

$head\ (x \# S) = x$   
 ⟨proof⟩

**theorem** *nonempty-tail* [simp]:

$tail\ (x \# S) = S$   
 ⟨proof⟩

**definition** (in *graph*)

$reach\ x \equiv \{y . (x, y) \in next^* \wedge y \neq nil\}$

**theorem** (in *graph*) *reach-nil* [simp]:  $reach\ nil = \{\}$

⟨proof⟩

**theorem** (in *graph*) *reach-next*:  $b \in reach\ a \implies (b, c) \in next \implies c \in reach\ a$

⟨proof⟩

**definition** (in *graph*)

```

    path S mrk ≡ {x . (∃ s . s ∈ S ∧ (s, x) ∈ next O (next ∩ ((-mrk)×(-mrk)))*)
  }
end
end

```

### 3 Marking Using a Set

```

theory SetMark
imports Graph DataRefinementIBP.DataRefinement
begin

```

We construct in this theory a diagram which computes all reachable nodes from a given root node in a graph. The graph is defined in the theory `Graph` and is given by a relation `next` on the nodes of the graph.

The diagram has only three ordered situation ( $init > loop > final$ ). The termination variant is a pair of a situation and a natural number with the lexicographic ordering. The idea of this ordering is that we can go from a bigger situation to a smaller one, however if we stay in the same situation the second component of the variant must decrease.

The idea of the algorithm is that it starts with a set  $X$  containing the root element and the root is marked. As long as  $X$  is not empty, if  $x \in X$  and  $y$  is an unmarked successor of  $x$  we add  $y$  to  $X$ . If  $x \in X$  has no unmarked successors it is removed from  $X$ . The algorithm terminates when  $X$  is empty.

```

datatype I = init | loop | final

```

```

declare I.split [split]

```

```

instantiation I :: well-founded-transitive
begin

```

**definition**

```

less-I-def: i < j ≡ (j = init ∧ (i = loop ∨ i = final)) ∨ (j = loop ∧ i = final)

```

**definition**

```

less-eq-I-def: (i::I) ≤ (j::I) ≡ i = j ∨ i < j

```

**instance**

```

⟨proof⟩

```

**end**

The set `path S mrk` contains all reachable nodes from  $S$  along paths with unmarked nodes.

**lemma** *trascI-less*:  $x \neq y \implies (a, x) \in R^* \implies$

$((a,x) \in (R \cap (-\{y\}) \times (-\{y\}))^* \vee (y,x) \in R \ O \ (R \cap (-\{y\}) \times (-\{y\}))^* )$   
 $\langle proof \rangle$

**lemma** (in graph) *add-set [simp]*:  $x \neq y \implies x \in path \ S \ mrk \implies x \in path \ (insert \ y \ S) \ (insert \ y \ mrk)$   
 $\langle proof \rangle$

**lemma** (in graph) *add-set2*:  $x \in path \ S \ mrk \implies x \notin path \ (insert \ y \ S) \ (insert \ y \ mrk) \implies x = y$   
 $\langle proof \rangle$

**lemma** (in graph) *del-stack [simp]*:  $(\forall \ y \ . \ (t, y) \in next \longrightarrow y \in mrk) \implies x \notin mrk \implies x \in path \ S \ mrk \implies x \in path \ (S - \{t\}) \ mrk$   
 $\langle proof \rangle$

**lemma** (in graph) *init-set [simp]*:  $x \in reach \ root \implies x \neq root \implies x \in path \ \{root\} \ \{root\}$   
 $\langle proof \rangle$

**lemma** (in graph) *init-set2*:  $x \in reach \ root \implies x \notin path \ \{root\} \ \{root\} \implies x = root$   
 $\langle proof \rangle$

### 3.1 Transitions

**definition** (in graph)

$Q1-a \equiv [ : X, mrk \rightsquigarrow X', mrk'. \ (root::'node) = nil \wedge X' = \{\} \wedge mrk' = mrk : ]$

**definition** (in graph)

$Q2-a \equiv [ : X, mrk \rightsquigarrow X', mrk' .$   
 $(root::'node) \neq nil \wedge X' = \{root::'node\} \wedge mrk' = \{root::'node\} : ]$

**definition** (in graph)

$Q3-a \equiv [ : X, mrk \rightsquigarrow X', mrk' .$   
 $(\exists \ x \in X \ . \ \exists \ y \ . \ (x, y) \in next \wedge y \notin mrk \wedge X' = X \cup \{y\} \wedge mrk' = mrk \cup \{y\}) : ]$

**definition** (in graph)

$Q4-a \equiv [ : X, mrk \rightsquigarrow X', mrk' .$   
 $(\exists \ x \in X \ . \ (\forall \ y \ . \ (x, y) \in next \longrightarrow y \in mrk) \wedge X' = X - \{x\} \wedge mrk' = mrk) : ]$

**definition** (in graph)

$Q5-a \equiv [ : X, mrk \rightsquigarrow X', mrk' . \ X = \{\} \wedge mrk = mrk' : ]$

### 3.2 Invariants

**definition** (in graph)

$Loop \equiv \{ (X, mrk) \ .$   
 $finite \ (-mrk) \wedge finite \ X \wedge X \subseteq mrk \wedge$

$$mrk \subseteq reach\ root \wedge reach\ root \cap -mrk \subseteq path\ X\ mrk\}$$

**definition**

$$trm \equiv \lambda (X, mrk) . 2 * card (-mrk) + card X$$

**definition**

$$term\text{-}eq\ t\ w = \{s . t\ s = w\}$$

**definition**

$$term\text{-}less\ t\ w = \{s . t\ s < w\}$$

**lemma** *union-term-eq [simp]*:  $(\bigcup w . term\text{-}eq\ t\ w) = UNIV$   
*<proof>*

**lemma** *union-less-term-eq [simp]*:  $(\bigcup v \in \{v . v < w\} . term\text{-}eq\ t\ v) = term\text{-}less\ t\ w$   
*<proof>*

**definition** (*in graph*)

$$Init \equiv \{ (X :: ('node\ set), mrk :: ('node\ set)) . finite (-mrk) \wedge mrk = \{\}\}$$

**definition** (*in graph*)

$$Final \equiv \{ (X :: ('node\ set), mrk :: ('node\ set)) . mrk = reach\ root\}$$

**definition** (*in graph*)

$$\begin{aligned} SetMarkInv\ i &= (case\ i\ of \\ I.init &\Rightarrow Init\ | \\ I.loop &\Rightarrow Loop\ | \\ I.final &\Rightarrow Final) \end{aligned}$$

**definition** (*in graph*)

$$\begin{aligned} SetMarkInvFinal\ i &= (case\ i\ of \\ I.final &\Rightarrow Final\ | \\ - &\Rightarrow \{\}) \end{aligned}$$

**definition** (*in graph*) [*simp*]:

$$\begin{aligned} SetMarkInvTerm\ w\ i &= (case\ i\ of \\ I.init &\Rightarrow Init\ | \\ I.loop &\Rightarrow Loop \cap \{s . trm\ s = w\}\ | \\ I.final &\Rightarrow Final) \end{aligned}$$

### 3.3 Diagram

**definition** (*in graph*)

$$\begin{aligned} SetMark &\equiv \lambda (i, j) . (case\ (i, j)\ of \\ (I.init, I.loop) &\Rightarrow Q1\text{-}a \sqcap Q2\text{-}a\ | \\ (I.loop, I.loop) &\Rightarrow Q3\text{-}a \sqcap Q4\text{-}a\ | \\ (I.loop, I.final) &\Rightarrow Q5\text{-}a\ | \\ - &\Rightarrow top) \end{aligned}$$

**lemma** (in graph) *SetMark-dmono* [simp]:  
*dmono SetMark*  
 ⟨proof⟩

### 3.4 Correctness of the transitions

**lemma** (in graph) *init-loop-1-a*[simp]:  $\models \text{Init } \{| Q1-a |\} \text{ Loop}$   
 ⟨proof⟩

**lemma** (in graph) *init-loop-2-a*[simp]:  $\models \text{Init } \{| Q2-a |\} \text{ Loop}$   
 ⟨proof⟩

**lemma** (in graph) *loop-loop-1-a* [simp]:  $\models (\text{Loop} \cap \{s . \text{trm } s = w\}) \{| Q3-a |\}$   
 $(\text{Loop} \cap \{s . \text{trm } s < w\})$   
 ⟨proof⟩

**lemma** (in graph) *loop-loop-2-a*[simp]:  $\models (\text{Loop} \cap \{s . \text{trm } s = w\}) \{| Q4-a |\}$   
 $(\text{Loop} \cap \{s . \text{trm } s < w\})$   
 ⟨proof⟩

**lemma** (in graph) *loop-final-a* [simp]:  $\models (\text{Loop} \cap \{s . \text{trm } s = w\}) \{| Q5-a |\}$   
*Final*  
 ⟨proof⟩

**lemma** *union-term-w*[simp]:  $(\bigcup w . \{s . t s = w\}) = \text{UNIV}$   
 ⟨proof⟩

**lemma** *union-less-term-w*[simp]:  $(\bigcup v \in \{v . v < w\} . \{s . t s = v\}) = \{s . t s < w\}$   
 ⟨proof⟩

**lemma** *sup-union*[simp]:  $\text{Sup } (\text{range } A) i = (\bigcup w . A w i)$   
 ⟨proof⟩

**lemma** *forall-simp* [simp]:  $(\forall a b . \forall x \in A . (a = (t x)) \longrightarrow (h x) \vee b \neq u x) = (\forall x \in A . h x)$   
 ⟨proof⟩

**lemma** *forall-simp2* [simp]:  $(\forall a b . \forall x \in A . \forall y . (a = t x y) \longrightarrow (h x y) \longrightarrow (g x y) \vee b \neq u x y) = (\forall x \in A . \forall y . h x y \longrightarrow g x y)$   
 ⟨proof⟩

### 3.5 Diagram correctness

The termination ordering for the *SetMark* diagram is the lexicographic ordering on pairs  $(i, n)$  where  $i \in I$  and  $n \in \text{nat}$ .

**interpretation** *DiagramTermination*  $\lambda (n :: \text{nat}) (i :: I) . (i, n)$   
 ⟨proof⟩

**theorem** (in graph) *SetMark-correct*:



$\models \text{SetMarkInv } \{|pt \text{ SetMark}|\} \text{ SetMarkInvFinal}$   
 $\langle \text{proof} \rangle$

**theorem** (in *graph*) *SetMark-correct1* [*simp*]:  
 $\text{Hoare-dgr } \text{SetMarkInv } \text{SetMark} (\text{SetMarkInv} \sqcap (- \text{grd } (\text{step } \text{SetMark})))$   
 $\langle \text{proof} \rangle$

**theorem** (in *graph*) *stack-not-nil* [*simp*]:  
 $(\text{mrk}, S) \in \text{Loop} \implies x \in S \implies x \neq \text{nil}$   
 $\langle \text{proof} \rangle$

**end**

## 4 Marking Using a Stack

**theory** *StackMark*  
**imports** *SetMark DataRefinementIBP.DataRefinement*  
**begin**

In this theory we refine the set marking diagram to a diagram in which the set is replaced by a list (stack). Iniatially the list contains the root element and as long as the list is nonempty and the top of the list has an unmarked successor  $y$ , then  $y$  is added to the top of the list. If the top does not have unmarked sucessors, it is removed from the list. The diagram terminates when the list is empty.

The data refinement relation of the two diagrams is true if the list has distinct elements and the elements of the list and the set are the same.

### 4.1 Transitions

**definition** (in *graph*)  
 $Q1'-a \equiv [\lambda (stk::('node \text{ list}), \text{mrk}::('node \text{ set})) . \{(stk::('node \text{ list}), \text{mrk}') .$   
 $\text{root} = \text{nil} \wedge \text{stk}' = [] \wedge \text{mrk}' = \text{mrk}\}:]$

**definition** (in *graph*)  
 $Q2'-a \equiv [\lambda (stk::('node \text{ list}), \text{mrk}::('node \text{ set})) . \{(stk', \text{mrk}') .$   
 $\text{root} \neq \text{nil} \wedge \text{stk}' = [\text{root}] \wedge \text{mrk}' = \text{mrk} \cup \{\text{root}\}\}:]$

**definition** (in *graph*)  
 $Q3'-a \equiv [\lambda (stk, \text{mrk}) . \{(stk', \text{mrk}') . \text{stk} \neq [] \wedge (\exists y . (\text{hd } \text{stk}, y) \in \text{next} \wedge$   
 $y \notin \text{mrk} \wedge \text{stk}' = y \# \text{stk} \wedge \text{mrk}' = \text{mrk} \cup \{y\})\}:]$

**definition** (in *graph*)  
 $Q4'-a \equiv [\lambda (stk, \text{mrk}) . \{(stk', \text{mrk}') . \text{stk} \neq [] \wedge$   
 $(\forall y . (\text{hd } \text{stk}, y) \in \text{next} \longrightarrow y \in \text{mrk}) \wedge \text{stk}' = \text{tl } \text{stk} \wedge \text{mrk}' = \text{mrk}\}:]$

**definition**  
 $Q5'-a \equiv [\lambda (stk, \text{mrk}) . \{(stk', \text{mrk}') . \text{stk} = [] \wedge \text{mrk}' = \text{mrk}\}:]$

## 4.2 Invariants

**definition**

$$Init' \equiv UNIV$$

**definition**

$$Loop' \equiv \{ (stk, mrk) . distinct\ stk \}$$

**definition**

$$Final' \equiv UNIV$$

**definition** [*simp*]:

$$\begin{aligned} StackMarkInv\ i &= (case\ i\ of \\ &I.init \Rightarrow Init' \mid \\ &I.loop \Rightarrow Loop' \mid \\ &I.final \Rightarrow Final') \end{aligned}$$

## 4.3 Data refinement relations

**definition**

$$R1-a \equiv \{ : stk, mrk \rightsquigarrow X, mrk' . mrk' = mrk : \}$$

**definition**

$$R2-a \equiv \{ : stk, mrk \rightsquigarrow X, mrk' . X = set\ stk \wedge (stk, mrk) \in Loop' \wedge mrk' = mrk : \}$$

**lemma** [*simp*]:  $R1-a \in Apply.Disjunctive$   
*<proof>*

**lemma** [*simp*]:  $R2-a \in Apply.Disjunctive$  *<proof>*

**definition** [*simp*]:

$$\begin{aligned} R-a\ i &= (case\ i\ of \\ &I.init \Rightarrow R1-a \mid \\ &I.loop \Rightarrow R2-a \mid \\ &I.final \Rightarrow R1-a) \end{aligned}$$

**lemma** [*simp*]: *Disjunctive-fun*  $R-a$  *<proof>*

**definition**

$$angelic-fun\ r = (\lambda\ i . \{ : r\ i : \})$$

**definition** (*in graph*)

$$\begin{aligned} StackMark-a &= (\lambda\ (i, j) . (case\ (i, j)\ of \\ &(I.init, I.loop) \Rightarrow Q1'-a \sqcap Q2'-a \mid \\ &(I.loop, I.loop) \Rightarrow Q3'-a \sqcap Q4'-a \mid \\ &(I.loop, I.final) \Rightarrow Q5'-a \mid \\ &- \Rightarrow \top)) \end{aligned}$$

#### 4.4 Data refinement of the transitions

**theorem** (in *graph*) *init-nil* [*simp*]:  
 *DataRefinement* ( $\{.Init.\} \circ Q1-a$ ) *R1-a* *R2-a* *Q1'-a*  
  $\langle proof \rangle$

**theorem** (in *graph*) *init-root* [*simp*]:  
 *DataRefinement* ( $\{.Init.\} \circ Q2-a$ ) *R1-a* *R2-a* *Q2'-a*  
  $\langle proof \rangle$

**theorem** (in *graph*) *step1* [*simp*]:  
 *DataRefinement* ( $\{.Loop.\} \circ Q3-a$ ) *R2-a* *R2-a* *Q3'-a*  
  $\langle proof \rangle$

**theorem** (in *graph*) *step2* [*simp*]:  
 *DataRefinement* ( $\{.Loop.\} \circ Q4-a$ ) *R2-a* *R2-a* *Q4'-a*  
  $\langle proof \rangle$

**theorem** (in *graph*) *final* [*simp*]:  
 *DataRefinement* ( $\{.Loop.\} \circ Q5-a$ ) *R2-a* *R1-a* *Q5'-a*  
  $\langle proof \rangle$

#### 4.5 Diagram data refinement

**lemma** *assert-comp-choice*:  $\{.p.\} \circ (S \sqcap T) = (\{.p.\} \circ S) \sqcap (\{.p.\} \circ T)$   
  $\langle proof \rangle$

**theorem** (in *graph*) *StackMark-DataRefinement* [*simp*]:  
 *DgrDataRefinement2* *SetMarkInv* *SetMark* *R-a* *StackMark-a*  
  $\langle proof \rangle$

#### 4.6 Diagram correctness

**theorem** (in *graph*) *StackMark-correct*:  
 *Hoare-dgr* (*R-a* .. *SetMarkInv*) *StackMark-a* ( $((R-a .. SetMarkInv) \sqcap (- \text{grd } (step (StackMark-a))))$ )  
  $\langle proof \rangle$

**end**

## 5 Generalization of Deutsch-Schorr-Waite Algorithm

**theory** *LinkMark*  
**imports** *StackMark*  
**begin**

In the third step the stack diagram is refined to a diagram where no extra memory is used. The relation *next* is replaced by two new variables *link*

and  $label$ . The variable  $label : node \rightarrow index$  associates a label to every node and the variable  $link : index \rightarrow node \rightarrow node$  is a collection of pointer functions indexed by the set  $index$  of labels. For  $x \in node$ ,  $link\ i\ x$  is the successor node of  $x$  along the function  $link\ i$ . In this context a node  $x$  is reachable if there exists a path from the root to  $x$  along the links  $link\ i$  such that all nodes in this path are not  $nil$  and they are labeled by a special label  $none \in index$ .

The stack variable  $S$  is replaced by two new variables  $p$  and  $t$  ranging over nodes. Variable  $p$  stores the head of  $S$ ,  $t$  stores the head of the tail of  $S$ , and the rest of  $S$  is stored by temporarily modifying the variables  $link$  and  $label$ .

This algorithm is a generalization of the Deutsch-Schorr-Waite graph marking algorithm because we have a collection of pointer functions instead of left and right only.

```

locale pointer = node +
  fixes none :: 'index
  fixes link0::'index  $\Rightarrow$  'node  $\Rightarrow$  'node
  fixes label0 :: 'node  $\Rightarrow$  'index

  assumes (nil::'node) = nil
begin
  definition next = {(a, b) . ( $\exists i . link0\ i\ a = b$ )  $\wedge$   $a \neq nil \wedge b \neq nil \wedge label0\ a = none$ }
end

sublocale pointer  $\subseteq$  link?: graph nil root next
  <proof>

```

The locale pointer fixes the initial values for the variables  $link$  and  $label$  and it defines the relation next as the union of all  $link\ i$  functions, excluding the mappings to  $nil$ , the mappings from  $nil$  as well as the mappings from elements which are not labeled by  $none$ .

The next two recursive functions,  $label\_0$ ,  $link\_0$  are used to compute the initial values of the variables  $label$  and  $link$  from their current values.

```

context pointer
begin
primrec
  label-0:: ('node  $\Rightarrow$  'index)  $\Rightarrow$  ('node list)  $\Rightarrow$  ('node  $\Rightarrow$  'index) where
    label-0 lbl [] = lbl |
    label-0 lbl (x # l) = label-0 (lbl(x := none)) l

lemma label-cong [cong]:  $f = g \implies xs = ys \implies pointer.label-0\ n\ f\ xs = pointer.label-0\ n\ g\ ys$ 
  <proof>

```

**primrec**

$link-0:: ('index \Rightarrow 'node \Rightarrow 'node) \Rightarrow ('node \Rightarrow 'index) \Rightarrow 'node \Rightarrow ('node\ list) \Rightarrow ('index \Rightarrow 'node \Rightarrow 'node) \mathbf{where}$   
 $link-0\ lnk\ lbl\ p\ [] = lnk\ |$   
 $link-0\ lnk\ lbl\ p\ (x\ \#\ l) = link-0\ (lnk((lbl\ x)\ :=\ ((lnk\ (lbl\ x))(x\ :=\ p))))\ lbl\ x\ l$

The function *stack* defined bellow is the main data refinement relation connecting the stack from the abstract algorithm to its concrete representation by temporarily modifying the variable *link* and *label*.

**primrec**

$stack:: ('index \Rightarrow 'node \Rightarrow 'node) \Rightarrow ('node \Rightarrow 'index) \Rightarrow 'node \Rightarrow ('node\ list) \Rightarrow bool \mathbf{where}$   
 $stack\ lnk\ lbl\ x\ [] = (x = nil) \ |$   
 $stack\ lnk\ lbl\ x\ (y\ \#\ l) =$   
 $(x \neq nil \wedge x = y \wedge \neg x \in set\ l \wedge stack\ lnk\ lbl\ (lnk\ (lbl\ x)\ x)\ l)$

**lemma** *label-out-range0* [simp]:

$\neg x \in set\ S \Longrightarrow label-0\ lbl\ S\ x = lbl\ x$   
 $\langle proof \rangle$

**lemma** *link-out-range0* [simp]:

$\neg x \in set\ S \Longrightarrow link-0\ link\ label\ p\ S\ i\ x = link\ i\ x$   
 $\langle proof \rangle$

**lemma** *link-out-range* [simp]:  $\neg x \in set\ S \Longrightarrow link-0\ link\ (label(x := y))\ p\ S = link-0\ link\ label\ p\ S$ 

$\langle proof \rangle$

**lemma** *empty-stack* [simp]:  $stack\ link\ label\ nil\ S = (S = [])$ 

$\langle proof \rangle$

**lemma** *stack-out-link-range* [simp]:  $\neg p \in set\ S \Longrightarrow stack\ (link(i := (link\ i)(p := q)))\ label\ x\ S = stack\ link\ label\ x\ S$ 

$\langle proof \rangle$

**lemma** *stack-out-label-range* [simp]:  $\neg p \in set\ S \Longrightarrow stack\ link\ (label(p := q))\ x\ S = stack\ link\ label\ x\ S$ 

$\langle proof \rangle$

**definition**

$g\ mrk\ lbl\ ptr\ x \equiv ptr\ x \neq nil \wedge ptr\ x \notin mrk \wedge lbl\ x = none$

**lemma** *g-cong* [cong]:  $mrk = mrk1 \Longrightarrow lbl = lbl1 \Longrightarrow ptr = ptr1 \Longrightarrow x = x1 \Longrightarrow$ 

$pointer.g\ n\ m\ mrk\ lbl\ ptr\ x = pointer.g\ n\ m\ mrk1\ lbl1\ ptr1\ x1$

$\langle proof \rangle$

## 5.1 Transitions

### definition

$$Q1''-a \equiv [ : p, t, lnk, lbl, mrk \rightsquigarrow p', t', lnk', lbl', mrk' . \\ root = nil \wedge p' = nil \wedge t' = nil \wedge lnk' = lnk \wedge lbl' = lbl \wedge mrk' = mrk : ]$$

### definition

$$Q2''-a \equiv [ : p, t, lnk, lbl, mrk \rightsquigarrow p', t', lnk', lbl', mrk' . \\ root \neq nil \wedge p' = root \wedge t' = nil \wedge lnk' = lnk \wedge lbl' = lbl \wedge mrk' = mrk \cup \\ \{root\} : ]$$

### definition

$$Q3''-a \equiv [ : p, t, lnk, lbl, mrk \rightsquigarrow p', t', lnk', lbl', mrk' . \\ p \neq nil \wedge \\ (\exists i . g \text{ mrk } lbl (lnk \ i) \ p \wedge \\ p' = lnk \ i \ p \wedge t' = p \wedge lnk' = lnk(i := (lnk \ i)(p := t)) \wedge lbl' = lbl(p \\ := i) \wedge \\ mrk' = mrk \cup \{lnk \ i \ p\} : ]$$

### definition

$$Q4''-a \equiv [ : p, t, lnk, lbl, mrk \rightsquigarrow p', t', lnk', lbl', mrk' . \\ p \neq nil \wedge \\ (\forall i . \neg g \text{ mrk } lbl (lnk \ i) \ p) \wedge t \neq nil \wedge \\ p' = t \wedge t' = lnk \ (lbl \ t) \ t \wedge lnk' = lnk(lbl \ t := (lnk \ (lbl \ t))(t := p)) \\ \wedge lbl' = lbl(t := none) \wedge mrk' = mrk : ]$$

### definition

$$Q5''-a \equiv [ : p, t, lnk, lbl, mrk \rightsquigarrow p', t', lnk', lbl', mrk' . \\ p \neq nil \wedge \\ (\forall i . \neg g \text{ mrk } lbl (lnk \ i) \ p) \wedge t = nil \wedge \\ p' = nil \wedge t' = t \wedge lnk' = lnk \wedge lbl' = lbl \wedge mrk' = mrk : ]$$

### definition

$$Q6''-a \equiv [ : p, t, lnk, lbl, mrk \rightsquigarrow p', t', lnk', lbl', mrk' . p = nil \wedge \\ p' = p \wedge t' = t \wedge lnk' = lnk \wedge lbl' = lbl \wedge mrk' = mrk : ]$$

## 5.2 Invariants

### definition

$$Init'' \equiv \{ (p, t, lnk, lbl, mrk) . lnk = link0 \wedge lbl = label0 \}$$

### definition

$$Loop'' \equiv UNIV$$

### definition

$$Final'' \equiv Init''$$

## 5.3 Data refinement relations

### definition

$R1'-a \equiv \{ : p, t, lnk, lbl, mrk \rightsquigarrow stk, mrk' . (p, t, lnk, lbl, mrk) \in \text{Init}'' \wedge mrk' = mrk : \}$

**definition**

$R2'-a \equiv \{ : p, t, lnk, lbl, mrk \rightsquigarrow stk, mrk' .$   
 $p = \text{head } stk \wedge$   
 $t = \text{head } (\text{tail } stk) \wedge$   
 $\text{stack } lnk \text{ lbl } t (\text{tail } stk) \wedge$   
 $\text{link}0 = \text{link-}0 \text{ lnk } lbl \text{ } p (\text{tail } stk) \wedge$   
 $\text{label}0 = \text{label-}0 \text{ lbl } (\text{tail } stk) \wedge$   
 $\neg \text{nil} \in \text{set } stk \wedge$   
 $mrk' = mrk : \}$

**lemma** [simp]:  $R1'-a \in \text{Apply.Disjunctive} \langle \text{proof} \rangle$

**lemma** [simp]:  $R2'-a \in \text{Apply.Disjunctive} \langle \text{proof} \rangle$

**definition** [simp]:

$R'-a \text{ } i = (\text{case } i \text{ of}$   
 $I.\text{init} \Rightarrow R1'-a \mid$   
 $I.\text{loop} \Rightarrow R2'-a \mid$   
 $I.\text{final} \Rightarrow R1'-a)$

**lemma** [simp]:  $\text{Disjunctive-fun } R'-a \langle \text{proof} \rangle$

## 5.4 Diagram

**definition**

$\text{LinkMark} = (\lambda (i, j) . (\text{case } (i, j) \text{ of}$   
 $(I.\text{init}, I.\text{loop}) \Rightarrow Q1''-a \sqcap Q2''-a \mid$   
 $(I.\text{loop}, I.\text{loop}) \Rightarrow Q3''-a \sqcap (Q4''-a \sqcap Q5''-a) \mid$   
 $(I.\text{loop}, I.\text{final}) \Rightarrow Q6''-a \mid$   
 $- \Rightarrow \top))$

**definition** [simp]:

$\text{LinkMarkInv } i = (\text{case } i \text{ of}$   
 $I.\text{init} \Rightarrow \text{Init}'' \mid$   
 $I.\text{loop} \Rightarrow \text{Loop}'' \mid$   
 $I.\text{final} \Rightarrow \text{Final}'')$

## 5.5 Data refinement of the transitions

**theorem**  $\text{init1-}a$  [simp]:

$\text{DataRefinement } (\{.\text{Init}'\} \circ Q1'-a) R1'-a R2'-a Q1''-a$   
 $\langle \text{proof} \rangle$

**theorem**  $\text{init2-}a$  [simp]:

$\text{DataRefinement } (\{.\text{Init}'\} \circ Q2'-a) R1'-a R2'-a Q2''-a$   
 $\langle \text{proof} \rangle$

**theorem** *step1-a* [*simp*]:  
*DataRefinement* (*{.Loop'.*) *o Q3'-a R2'-a R2'-a Q3''-a*  
*<proof>*

**lemma** *neqif* [*simp*]:  $x \neq y \implies (\text{if } y = x \text{ then } a \text{ else } b) = b$   
*<proof>*

**theorem** *step2-a* [*simp*]:  
*DataRefinement* (*{.Loop'.*) *o Q4'-a R2'-a R2'-a Q4''-a*  
*<proof>*

**lemma** *setsimp*:  $a = c \implies (x \in a) = (x \in c)$   
*<proof>*

**theorem** *step3-a* [*simp*]:  
*DataRefinement* (*{.Loop'.*) *o Q4'-a R2'-a R2'-a Q5''-a*  
*<proof>*

**theorem** *final-a* [*simp*]:  
*DataRefinement* (*{.Loop'.*) *o Q5'-a R2'-a R1'-a Q6''-a*  
*<proof>*

## 5.6 Diagram data refinement

**lemma** *apply-fun-index* [*simp*]:  $(r \dots P) i = (r i) (P i)$  *<proof>*

**lemma** [*simp*]: *Disjunctive-fun* ( $r :: ('c \implies 'a :: \text{complete-lattice} \implies 'b :: \text{complete-lattice})$ )

$\implies \text{mono-fun } r$   
*<proof>*

**theorem** *LinkMark-DataRefinement-a* [*simp*]:  
*DgrDataRefinement2* (*R-a .. SetMarkInv*) *StackMark-a R'-a LinkMark*  
*<proof>*

**lemma** [*simp*]: *mono Q1'-a* *<proof>*

**lemma** [*simp*]: *mono Q2'-a* *<proof>*

**lemma** [*simp*]: *mono Q3'-a* *<proof>*

**lemma** [*simp*]: *mono Q4'-a* *<proof>*

**lemma** [*simp*]: *mono Q5'-a* *<proof>*

**lemma** [*simp*]: *dmono StackMark-a*  
*<proof>*

## 5.7 Diagram correctness

**theorem** *LinkMark-correct*:



```

    Hoare-dgr (R'-a .. (R-a .. SetMarkInv)) LinkMark ((R'-a .. (R-a .. SetMarkInv))
  □ (- grd (step LinkMark)))
  ⟨proof⟩

end
end

```

## 6 Deutsch-Schorr-Waite Marking Algorithm

```

theory DSWMark
imports LinkMark
begin

```

Finally, we construct the Deutsch-Schorr-Waite marking algorithm by assuming that there are only two pointers (*left*, *right*) from every node. There is also a new variable, *atom* : *node* → *bool* which associates to every node a Boolean value. The data invariant of this refinement step requires that *index* has exactly two distinct elements *none* and *some*, *left* = *link none*, *right* = *link some*, and *atom* *x* is true if and only if *label* *x* = *some*.

We use a new locale which fixes the initial values of the variables *left*, *right*, and *atom* in *left0*, *right0*, and *atom0* respectively.

```

datatype Index = none | some

```

```

locale classical = node +
  fixes left0 :: 'node ⇒ 'node
  fixes right0 :: 'node ⇒ 'node
  fixes atom0 :: 'node ⇒ bool
  assumes (nil::'node) = nil
  begin
    definition
      link0 i = (if i = (none::Index) then left0 else right0)

    definition
      label0 x = (if atom0 x then (some::Index) else none)
  end

```

```

sublocale classical ⊆ dsw?: pointer nil root none::Index link0 label0
  ⟨proof⟩

```

```

context classical
begin

```

```

lemma [simp]:
  (label0 = (λ x . if atom x then some else none)) = (atom0 = atom)
  ⟨proof⟩

```

```

definition
  gg mrk atom ptr x ≡ ptr x ≠ nil ∧ ptr x ∉ mrk ∧ ¬ atom x

```

## 6.1 Transitions

### definition

$QQ1-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $root = nil \wedge p' = nil \wedge t' = nil \wedge mrk' = mrk \wedge left' = left$   
 $\wedge right' = right \wedge atom' = atom : ]$

### definition

$QQ2-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $root \neq nil \wedge p' = root \wedge t' = nil \wedge mrk' = mrk \cup \{root\}$   
 $\wedge left' = left \wedge right' = right \wedge atom' = atom : ]$

### definition

$QQ3-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $p \neq nil \wedge gg\ mrk\ atom\ left\ p \wedge$   
 $p' = left\ p \wedge t' = p \wedge mrk' = mrk \cup \{left\ p\} \wedge$   
 $left' = left(p := t) \wedge right' = right \wedge atom' = atom : ]$

### definition

$QQ4-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $p \neq nil \wedge gg\ mrk\ atom\ right\ p \wedge$   
 $p' = right\ p \wedge t' = p \wedge mrk' = mrk \cup \{right\ p\} \wedge$   
 $left' = left \wedge right' = right(p := t) \wedge atom' = atom(p := True) : ]$

### definition

$QQ5-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $p \neq nil \wedge \text{--- not needed in the proof}$   
 $\neg gg\ mrk\ atom\ left\ p \wedge \neg gg\ mrk\ atom\ right\ p \wedge$   
 $t \neq nil \wedge \neg atom\ t \wedge$   
 $p' = t \wedge t' = left\ t \wedge mrk' = mrk \wedge$   
 $left' = left(t := p) \wedge right' = right \wedge atom' = atom : ]$

### definition

$QQ6-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $p \neq nil \wedge \text{--- not needed in the proof}$   
 $\neg gg\ mrk\ atom\ left\ p \wedge \neg gg\ mrk\ atom\ right\ p \wedge$   
 $t \neq nil \wedge atom\ t \wedge$   
 $p' = t \wedge t' = right\ t \wedge mrk' = mrk \wedge$   
 $left' = left \wedge right' = right(t := p) \wedge atom' = atom(t := False) : ]$

### definition

$QQ7-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $p \neq nil \wedge$   
 $\neg gg\ mrk\ atom\ left\ p \wedge \neg gg\ mrk\ atom\ right\ p \wedge$   
 $t = nil \wedge$   
 $p' = nil \wedge t' = t \wedge mrk' = mrk \wedge$   
 $left' = left \wedge right' = right \wedge atom' = atom : ]$

### definition

$QQ8-a \equiv [ : p, t, left, right, atom, mrk \rightsquigarrow p', t', left', right', atom', mrk' .$   
 $p = nil \wedge p' = p \wedge t' = t \wedge mrk' = mrk \wedge left' = left \wedge right' = right \wedge atom'$   
 $= atom:]$

## 7 Data refinement relation

### definition

$RR-a \equiv \{ : p, t, left, right, atom, mrk \rightsquigarrow p', t', lnk, lbl, mrk' .$   
 $lnk\ none = left \wedge lnk\ some = right \wedge$   
 $lbl = (\lambda x . \text{if } atom\ x \text{ then } some \text{ else } none) \wedge$   
 $p' = p \wedge t' = t \wedge mrk' = mrk : \}$

### definition [simp]:

$R''-a\ i = RR-a$

### definition

$ClassicMark = (\lambda (i, j) . (\text{case } (i, j) \text{ of}$   
 $(I.init, I.loop) \Rightarrow QQ1-a \sqcap QQ2-a \mid$   
 $(I.loop, I.loop) \Rightarrow (QQ3-a \sqcap QQ4-a) \sqcap ((QQ5-a \sqcap QQ6-a) \sqcap QQ7-a) \mid$   
 $(I.loop, I.final) \Rightarrow QQ8-a \mid$   
 $- \Rightarrow \top))$

### 7.1 Data refinement of the transitions

#### theorem *init1-a* [simp]:

$DataRefinement (\{.Init''.\} \circ Q1''-a) RR-a RR-a QQ1-a$   
 $\langle proof \rangle$

#### theorem *init2-a* [simp]:

$DataRefinement (\{.Init''.\} \circ Q2''-a) RR-a RR-a QQ2-a$   
 $\langle proof \rangle$

#### lemma *index-simp*:

$(u = v) = (u\ none = v\ none \wedge u\ some = v\ some)$   
 $\langle proof \rangle$

#### theorem *step1-a* [simp]:

$DataRefinement (\{.Loop''.\} \circ Q3''-a) RR-a RR-a QQ3-a$   
 $\langle proof \rangle$

#### theorem *step2-a*[simp]:

$DataRefinement (\{.Loop''.\} \circ Q3''-a) RR-a RR-a QQ4-a$   
 $\langle proof \rangle$

#### theorem *step3-a* [simp]:

$DataRefinement (\{.Loop''.\} \circ Q4''-a) RR-a RR-a QQ5-a$   
 $\langle proof \rangle$

**lemma** *if-set-elim*:  $(x \in (\text{if } b \text{ then } A \text{ else } B)) = ((b \wedge x \in A) \vee (\neg b \wedge x \in B))$   
 ⟨proof⟩

**theorem** *step4-a* [simp]:  
 DataRefinement ( $\{.Loop''.\}$  o  $Q4''-a$ ) RR-a RR-a QQ6-a  
 ⟨proof⟩

**theorem** *step5-a* [simp]:  
 DataRefinement ( $\{.Loop''.\}$  o  $Q5''-a$ ) RR-a RR-a QQ7-a  
 ⟨proof⟩

**theorem** *final-step-a* [simp]:  
 DataRefinement ( $\{.Loop''.\}$  o  $Q6''-a$ ) RR-a RR-a QQ8-a  
 ⟨proof⟩

## 7.2 Diagram data refinement

**lemma** [simp]: *mono RR-a* ⟨proof⟩  
**lemma** [simp]: *RR-a ∈ Apply.Disjunctive* ⟨proof⟩  
**lemma** [simp]: *Disjunctive-fun R''-a* ⟨proof⟩

**lemma** [simp]: *mono-fun R''-a* ⟨proof⟩

**lemma** [simp]: *mono Q1''-a* ⟨proof⟩  
**lemma** [simp]: *mono Q2''-a* ⟨proof⟩  
**lemma** [simp]: *mono Q3''-a* ⟨proof⟩  
**lemma** [simp]: *mono Q4''-a* ⟨proof⟩  
**lemma** [simp]: *mono Q5''-a* ⟨proof⟩  
**lemma** [simp]: *mono Q6''-a* ⟨proof⟩

**lemma** [simp]: *dmono LinkMark*  
 ⟨proof⟩

**theorem** *ClassicMark-DataRefinement-a* [simp]:  
 DgrDataRefinement2 ( $R'-a \dots (R-a \dots \text{SetMarkInv})$ ) LinkMark  $R''-a$  ClassicMark  
 ⟨proof⟩

## 7.3 Diagram corectness

**theorem** *ClassicMark-correct-a* [simp]:  
 Hoare-dgr ( $R''-a \dots (R'-a \dots (R-a \dots \text{SetMarkInv}))$ ) ClassicMark  
 ( $(R''-a \dots (R'-a \dots (R-a \dots \text{SetMarkInv})) \sqcap (\neg \text{grd} (\text{step ClassicMark})))$ )  
 ⟨proof⟩

We have proved the correctness of the final algorithm, but the pre and the post conditions involve the angelic choice operator and they depend on all data refinement steps we have used to prove the final diagram. We simplify these conditions and we show that we obtained indeed the corectness of the marking algorithm.

The predicate *ClassicInit* which is true for the *init* situation states that initially the variables *left*, *right*, and *atom* are equal to their initial values and also that no node is marked.

The predicate *ClassicFinal* which is true for the *final* situation states that at the end the values of the variables *left*, *right*, and *atom* are again equal to their initial values and the variable *mrk* records all reachable nodes. The reachable nodes are defined using our initial *next* relation, however if we unfold all locale interpretations and definitions we see easily that a node *x* is reachable if there is a path from *root* to *x* along *left* and *right* functions, and all nodes in this path have the *atom* bit false.

**definition**

$$\begin{aligned} \textit{ClassicInit} &= \{(p, t, \textit{left}, \textit{right}, \textit{atom}, \textit{mrk}) . \\ &\quad \textit{atom} = \textit{atom0} \wedge \textit{left} = \textit{left0} \wedge \textit{right} = \textit{right0} \wedge \\ &\quad \textit{finite} (\neg \textit{mrk}) \wedge \textit{mrk} = \{\}\} \end{aligned}$$

**definition**

$$\begin{aligned} \textit{ClassicFinal} &= \{(p, t, \textit{left}, \textit{right}, \textit{atom}, \textit{mrk}) . \\ &\quad \textit{atom} = \textit{atom0} \wedge \textit{left} = \textit{left0} \wedge \textit{right} = \textit{right0} \wedge \\ &\quad \textit{mrk} = \textit{reach root}\} \end{aligned}$$

**theorem** [*simp*]:

$$\begin{aligned} \textit{ClassicInit} &\subseteq (\textit{RR-a} (\textit{R1'-a} (\textit{R1-a} (\textit{SetMarkInv init})))) \\ &\langle \textit{proof} \rangle \end{aligned}$$

**theorem** [*simp*]:

$$\begin{aligned} (\textit{RR-a} (\textit{R1'-a} (\textit{R1-a} (\textit{SetMarkInv final})))) &\leq \textit{ClassicFinal} \\ &\langle \textit{proof} \rangle \end{aligned}$$

The indexed predicate *ClassicPre* is the precondition of the diagram, and since we are only interested in starting the marking diagram in the *init* situation we set *ClassicPre loop* = *ClassicPre final* =  $\emptyset$ .

**definition** [*simp*]:

$$\begin{aligned} \textit{ClassicPre } i &= (\textit{case } i \textit{ of} \\ &\quad \textit{I.init} \Rightarrow \textit{ClassicInit} \mid \\ &\quad \textit{I.loop} \Rightarrow \{\} \mid \\ &\quad \textit{I.final} \Rightarrow \{\}) \end{aligned}$$

We are interested on the other hand that the marking diagram terminates only in the *final* situation. In order to achieve this we define the postcondition of the diagram as the indexed predicate *ClassicPost* which is empty on every situation except *final*.

**definition** [*simp*]:

$$\begin{aligned} \textit{ClassicPost } i &= (\textit{case } i \textit{ of} \\ &\quad \textit{I.init} \Rightarrow \{\} \mid \\ &\quad \textit{I.loop} \Rightarrow \{\} \mid \\ &\quad \textit{I.final} \Rightarrow \textit{ClassicFinal}) \end{aligned}$$

**lemma** *exists-or*:  
 $(\exists x . p \ x \vee q \ x) = ((\exists x . p \ x) \vee (\exists x . q \ x))$   
 $\langle proof \rangle$

**lemma** [*simp*]:  
 $(- \text{grd} (\text{step } \text{ClassicMark})) \text{init} = \{\}$   
 $\langle proof \rangle$

**lemma** [*simp*]:  $\text{grd } \top = \perp$   
 $\langle proof \rangle$

**lemma** [*simp*]:  
 $(- \text{grd} (\text{step } \text{ClassicMark})) \text{loop} = \{\}$   
 $\langle proof \rangle$

The final theorem states the correctness of the marking diagram with respect to the precondition *ClassicPre* and the postcondition *ClassicPost*, that is, if the diagram starts in the initial situation, then it will terminate in the final situation, and it will mark all reachable nodes.

**lemma** [*simp*]: *mono QQ1-a*  $\langle proof \rangle$

**lemma** [*simp*]: *mono QQ2-a*  $\langle proof \rangle$

**lemma** [*simp*]: *mono QQ3-a*  $\langle proof \rangle$

**lemma** [*simp*]: *mono QQ4-a*  $\langle proof \rangle$

**lemma** [*simp*]: *mono QQ5-a*  $\langle proof \rangle$

**lemma** [*simp*]: *mono QQ6-a*  $\langle proof \rangle$

**lemma** [*simp*]: *mono QQ7-a*  $\langle proof \rangle$

**lemma** [*simp*]: *mono QQ8-a*  $\langle proof \rangle$

**lemma** [*simp*]: *dmono ClassicMark*  
 $\langle proof \rangle$

**theorem**  $\models \text{ClassicPre } \{ | \text{pt } \text{ClassicMark} \ | \} \text{ClassicPost}$   
 $\langle proof \rangle$

**end**

**end**

## References

- [1] J.-R. Abrial. Event based sequential program development: Application to constructing a pointer program. In K. Araki, S. Gnesi, and D. Mandrioli, editors, *FME*, volume 2805 of *Lecture Notes in Computer Science*, pages 51–74. Springer, 2003.
- [2] R. J. Back. *Correctness preserving program refinements: proof theory and applications*, volume 131 of *Mathematical Centre Tracts*. Mathe-

matisch Centrum, Amsterdam, 1980.

- [3] R.-J. Back. Semantic correctness of invariant based programs. In *International Workshop on Program Construction*, Chateau de Bonas, France, 1980.
- [4] R.-J. Back. Invariant based programs and their correctness. In W. Biermann, G. Guiho, and Y. Kodratoff, editors, *Automatic Program Construction Techniques*, pages 223–242. MacMillan Publishing Company, 1983.
- [5] R.-J. Back. Invariant based programming: Basic approach and teaching experience. *Formal Aspects of Computing*, 2008.
- [6] R.-J. Back and V. Preoteasa. Semantics and proof rules of invariant based programs. Technical Report 903, TUCS, Jul 2008.
- [7] R. J. Back and J. von Wright. Encoding, decoding and data refinement. *Formal Aspects of Computing*, 12:313–349, 2000.
- [8] W. DeRoever and K. Engelhardt. *Data Refinement: Model-Oriented Proof Methods and Their Comparison*. Cambridge University Press, New York, NY, USA, 1999.
- [9] C. A. R. Hoare. Proof of correctness of data representations. *Acta Informatica*, 1(4), Dec. 1972.
- [10] D. E. Knuth. *The art of computer programming, volume 1 (3rd ed.): fundamental algorithms*. Addison Wesley Longman Publishing Co., Inc., Redwood City, CA, USA, 1997.
- [11] F. Mehta and T. Nipkow. Proving pointer programs in higher-order logic. *Information and Computation*, 199:200–227, 2005.
- [12] V. Preoteasa and R.-J. Back. Data refinement of invariant based programs. *Electronic Notes in Theoretical Computer Science*, 259:143 – 163, 2009. Proceedings of the 14th BCS-FACS Refinement Workshop (REFINE 2009).
- [13] V. Preoteasa and R.-J. Back. Semantics and data refinement of invariant based programs. In G. Klein, T. Nipkow, and L. Paulson, editors, *Archive of Formal Proofs*. <http://isa-afp.org/entries/DataRefinementIBP.shtml>, May 2010. Formal proof development. Submitted.
- [14] H. Schorr and W. M. Waite. An efficient machine-independent procedure for garbage collection in various list structures. *Commun. ACM*, 10(8):501–506, 1967.